TEMPLATE = app
TARGET = dropcache-mdn

target.path = /opt/$${TARGET}/bin
INSTALLS *= target

QT += core gui
CONFIG += meegotouch

QMAKE_CXXFLAGS *= -fvisibility=hidden -fvisibility-inlines-hidden

#DEFINES *= QT_NO_CAST_TO_ASCII QT_NO_CAST_FROM_ASCII

# icons
svg_icon.files = $${TARGET}.svg
svg_icon.path = /opt/$${TARGET}/share/icons

png_icon.files = $${TARGET}.png
png_icon.path = /opt/$${TARGET}/share/icons

INSTALLS *= svg_icon png_icon

# scripts
scriptfile.files = *.sh
scriptfile.path = /opt/$${TARGET}/bin
INSTALLS *= scriptfile

# desktop file
desktopfile.files = $${TARGET}.desktop
desktopfile.path = /usr/share/applications
INSTALLS *= desktopfile

# theme files
themes.path = $$(DESTDIR)/usr/share/themes/base/meegotouch/$${TARGET}/style
themes.files = *.css

INSTALLS *= themes

# styles

STYLE_HEADERS *= \
    dropcachewidgetstyle.h

# source
SOURCES += main.cpp \
    mainpage.cpp \
    dropcachewidget.cpp

HEADERS += \
    mainpage.h \
    translations.h \
    dropcachewidgetstyle.h \
    dropcachewidget.h

OTHER_FILES += \
    dropcache-mdn.desktop \
    dropcache-mdn.svg \
    dropcache-mdn.png \
    dropcache-mdn.css \
    dropcache.sh
